Использование рекурсии при создании графических образов.

Рекурсивными называют процедуры (или функции), которые содержат обращение к самим себе.  Это обращение может быть прямое – вызовом процедуры внутри самой процедуры (PROC1),  или косвенное – вызовом других процедур,  внутри которых есть вызов исходной процедуры. При этом все процедуры в такой цепочке вызова будут рекурсивными (PROC2 – PROC3).

Рекурсия позволяет компактно реализовать некоторые алгоритмы.  При этом важно корректно определять условие прекращения работы рекурсивной процедуры во избежание зацикливания программы.  Приведем пример (см. файл vector1.pas):


Рекурсивное построения лабиринта последовательным построением отрезков прямых


uses Crt,Graph;  var Gd,Gm,dL,du: integer;

PROCEDURE VECTOR( L,ugol: integer);
      var dx,dy: Integer;

{ Отрезок длиной L строится под углом ugol (град.) из текущего положения курсора:
    рекурсия с построением на прямом проходе последовательности рекурсивных вызовов }

begin  { составляющие вектора по осям }
        dx:=round( L*cos( ugol*pi/180));
        dy:=round( L*sin( ugol*pi/180));

    LineRel(dx,dy);  delay(5000);

{ условие продолжения рекурсии }
  If (L>abs(dL))and(L<500) then
{ меняем длину отрезка на dL и направление на du }
 VECTOR(L+dL,ugol+du);  { рекурсивный вызов }   End;

begin  Gd:= 0;  InitGraph(Gd,Gm,'c:\bp\bgi');
       MoveTo(GetMaxX div 2,GetMaxY div 2);  { начальная позиция }

       dL:=-20;  du:=90;  { изменение длины и угла вектора }
       VECTOR(240,0);      ReadKey; CloseGraph end.

Поскольку внутри процедуры VECTOR происходит вызов этой же процедуры с новыми фактическими параметрами (x, y),  то последующий вектор начнется с конца предыдущего. Варьируя dL и du, можно строить различные спиралевидные узоры. Если рекурсивный вызов перенести до операторов расчета очередного смещения (dx, dy) с выводом LineRel (dx,dy), то построение пройдет на обратном проходе последовательности рекурсивных вызовов  (см. файл vector2.pas).
"Вышивальный автомат"

Следующий пример показывает использование рекурсии при построении узоров,  напоминающих кружева.  Задается размер "стежка" - короткой линии длиной dl, а также функция варьирования угла ugol, под которым этот стежок строится, в зависимости от номера шага  n.

uses Graph,Crt;
var gD,gM: integer;  L,ugol,base,a,b: double;
    c: char;  s1,s2: string;  stop,draw: boolean;

PROCEDURE STEP(x,y:double; N,Nmax: integer);
          begin   { N – счетчик стежков узора }
base:=2*pi*N/Nmax; { угол, определяющий базовую линию узора }
{ На базовую линию (здесь это окружность диаметром N*L/pi )
  накладываем периодическое отклонение с амплитудой a и частотой b }

ugol:=base+a*sin(b*base);

if N=0 then MoveTo(round(x),round(y));  { переход к стартовой точке }

x:=x+L*cos(ugol);
y:=y+L*sin(ugol);  LineTo(round(x),round(y)); { прорисовка стежка }

if N<Nmax then  STEP(x,y, N+1,Nmax)   end;

begin  gD:=0;  InitGraph(gD,gM,'');

       L:=3;            { размер "стежка" }
       a:=1; b:=2;      { начальные значения коэффициентов }
  STEP(200,170,0,400);  { вывод узора }
 stop:=false;
REPEAT  { модификация узора подбором величин a,b }

             c:=readkey;   { считывание с клавиатуры }
if c=#0 then c:=readkey;   { для курсорных клавиш }
     draw:=true;   { признак перерисовки }

CASE c OF          { анализируем код нажатой клавиши }

  #72: a:=a+0.1; { вверх  }    #80: a:=a-0.1; { вниз  }
  #77: b:=b+1;   { вправо }    #75: b:=b-1;   { влево }
  #27: stop:=true; { при нажатии Esc - останов программы }

 Else  draw:=false;    end;

if stop then Break;

If draw then begin  ClearDevice;
Str(a:5:1, s1);  Str(b:5:1, s2); { вывод текущих коэффициентов }
outtextxy(460,390,'a='+s1+'  вверх/вниз');
outtextxy(460,410,'b='+s2+'  <-/->');

     STEP(200,170,0,400) { вывод узора }  end
 until stop;   CloseGraph  end.


Приведенная программа позволяет строить весьма разнообразные кружевные узоры даже при взятой сравнительно простой зависимости варьирования направления "вышивания" ugol:=base+a*sin(b*base).  При усложнении зависимости, например, наложением на базовое направление двух пульсаций  ugol:=base+a1*sin(b1*base)+ a2*sin(b2*base  (см. файл step2.pas) разнообразие картин, как и их изящество существенно возрастает – нужно лишь удачно подобрать соотношение коэффициентов a1,b1,a2,b2. Полезно также ввести варьирование размера стежка и/или количества стежков – иногда размер узора получается слишком большим или слишком маленьким. Рост количества стежков при уменьшении их размера позволяет уменьшить угловатость узора.


В качестве базового направления необязательно брать окружность – можно любую линию, например, при base=const – прямую. Из отрезков прямых можно делать "заготовки" для кружевных рамок. Можно варьировать и размер стежка в зависимости от его номера (например, циклически).

Рекурсивное построение разветвляющихся наборов ломаных ("деревьев")


Если в процедуре проводить несколько рекурсивных вызовов, то можно создавать "древовидные структуры" – количество "отростков" в развилках будет равно количеству вызовов. Серьезные приложения подобных алгоритмов – в задачах по теории графов, рекурсивных структурах типа структуры каталогов MS-DOS. Графические построения "древовидных структур" кроме наглядной демонстрации методики позволяют просто получать красивые образы.

uses Graph,Crt;   var gD,gM: integer;

PROCEDURE BETKA(x,y,L: integer; a: double);

{ Рекурсивное рисование "веток" - последовательно расположенных отрезков  из точки (x,y), постепенно меняющих размер L и угол ориентации a }

Const  b=pi/4;  { угол между " ветками"    }
       m=0.8;   { коэффициент пошагового изменения размера "ветки" }
  var x0,y0: integer;

begin
{ Ограничение диапазона размеров "ветки" является условием прекращения рекурсии }
  if(L<9) or (L>100) then EXIT;
{ Установка стилей для разных элементов рисунка }
if L>40 then Setcolor( 6)
        else Setcolor(10);
if L<30 then SetLineStyle(0,0,1)
        else SetLineStyle(0,0,3);

Moveto(x,y);   { переход к стартовой точке рисования }
       x0:=x;  y0:=y; { запоминаем "начало ветки"}
x:=x+round(L*cos(a)); { координаты "конца ветки"}
y:=y+round(L*sin(a)); Lineto(x,y); { рисование "ветки" в виде отрезка }
BETKA(x,y,round(L*m),a-b/2);  { первый "отросток" - отклонение на -b/2 }
BETKA(x,y,round(L*m),a+b/2);  { второй "отросток" - отклонение на +b/2 }

 { возврат к предшествующим координатам перед началом новой ломаной }
  Moveto(x0,y0)  end;

begin  gD:=0; InitGraph(gD,gM,'');

BETKA(300,450,90,-pi/2); { в процедуру передаем параметры "базовой ветки"}

readkey;  CloseGraph  end.


Приведенный алгоритм нетрудно разнообразить, например, добавлением количества рекурсивных вызовов увеличить количество отростков (см. файл vetka2.pas), а постановкой ограничительных условий перед рекурсивными вызовами и варьированием коэффициента пошагового изменения размеров менять стиль получаемых образов.

Рекурсивное построение "сателлитов"

Вокруг базовой фигуры (здесь окружность) строится несколько ее копий уменьшенного размера, далее процесс рекурсивно повторяется для каждой из копий (см. файл satellit.pas).

uses Crt, Graph;
 var Gd,Gm,m,n: integer;
           x,y, r,r1, ks,kr: double;

PROCEDURE SAT( x,y,r,r1: double; n: integer);
      var x1,y1: double; i: Integer;

{ от фигуры размером r с центром  x,y   рекурсивно строится   m фигур-сателлитов
   размером r=r/ks  равномерно по периметру на расстоянии r1=kr*r }

begin
      If n=0 then EXIT;   { прерывание рекурсивного вызова }
         setcolor(15-n);  { цвет очередной рекурсивной группы }
         circle(round(x),round(y),round(r));   { вывод фигуры }
         r1:=kr*r;   { расстояние до фигур-сателлитов }

For i:=1 to m do begin      { рекурсивные вызовы }

   x1:= x+r1*cos(2*pi*i/m); { координаты центра i-го сателлита }
   y1:= y+r1*sin(2*pi*i/m);

      SAT( x1,y1,r/ks,r1,n-1)  end
End;

begin  Gd:=0;  InitGraph(Gd,Gm,'');
       x:=GetMaxX div 2; { центр начальной фигуры }
       y:=GetMaxY div 2;

      ks:=3;             { коэффициент уменьшения размера сателлитов }
      kr:=2;             { отношение радиуса орбиты к размеру фигуры }
       m:=6;             { число сателлитов }
       r:=90;            { размер начальной фигуры }
       n:=5;             { глубина рекурсии }
      SAT( x,y,r,r1,n);

readkey; CloseGraph end.

